

#### 2.1.4 Bi-factor analysis using ROTATION=BI-GEOMIN and direct effects second-order EFA (DSEFA) using PSEM

### EFA Variations

- Hypothesis about the number of factors:
  - ANALYSIS: TYPE = EFA
  - ESEM (\*1)
  - PSEM with GEOMIN priors
  - Second-order exploratory factor analysis (SEFA) using PSEM with GEOMIN priors
  - **Bi-factor analysis using ROTATION = BI-GEOMIN and direct second-order exploratory factor analysis (DSEFA) using PSEM**
- Hypothesis about the number of factors and key items:
  - ESEM with Target rotation
  - PSEM with ALF priors for cross loadings
- Comparing EFA methods
- Special models:
  - ESEM with PSEM priors for residual covariances
  - PSEM finding a small number of cross-loadings
  - Mixture EFA
  - Twolevel EFA

Slide 31 once again shows the outline of the various EFA Variations. We now turn to bi-factor analysis. Both old and new, improved techniques are discussed.

The focus is on the new method of direct second-order exploratory factor analysis (DSEFA), which uses PSEM.

## Input for Bi-Factor Confirmatory Factor Analysis Using 1 General and 4 Specific Factors for H&S19

```
MODEL:    ! CFA
          spatial BY visual-flags*;
          verbal BY general-wordm*;
          speed BY addition-straight*;
          memory BY wordr-figurew*;
          g BY visual-figurew*;
          spatial-g@1;
          ! uncorrelated factors:
          g WITH spatial-memory @0;
          spatial-memory WITH spatial-memory@0;
```

- CFA model fit with 1 + 4 factors, both schools (N=301) using MLR:  
Chi-square (133) = 242, p = 0.0000

Slide 32 shows the input for the bi-factor CFA model for the H&S19 data that we saw the model diagram for earlier. The 5 factors are specified to be uncorrelated in line with the diagram. The model fit is shown at the bottom.

## Exploratory Bi-Factor Analysis: BI-GEOMIN Rotation

---

### Bi-Factor EFA

ANALYSIS:

TYPE = EFA 5 5;  
ROTATION = BI-GEOMIN;

---

### Bi-Factor ESEM

ANALYSIS:

ROTATION = BI-GEOMIN;

MODEL:

fg f1-f4 by y1-y19(\*1);

---

- BI-GEOMIN uses the Jennrich & Bentler (2011, 2012) method
- Overview of bifactor analysis in Reise, Mansolf, Haviland (2023)
  - Waller (2018): Direct Schmid-Leiman
- New approach in Mplus 9.1: Direct effects second-order factor analysis (DSEFA) using PSEM priors (Asparouhov & Muthén, 2026)

Slide 33 shows input for exploratory bi-factor analysis using TYPE = EFA as well as using the equivalent ESEM specification. For ESEM, the general factor is called fg. Alternatively, this could be written as f1-f5 BY y1-y19(\*1).

Both approaches use the BI-GEOMIN rotation, which is based on the Jennrich-Bentler Gradient Projection Algorithm (the Psychometrika references are given in the Reference section).

A very helpful overview of bi-factor analysis, both confirmatory and exploratory, is given in the chapter by Reise and others in the SEM handbook of 2003 (see reference section). That chapter also refers to a paper by Waller which inspired the new DSEFA approach in Mplus discussed in Asparouhov- Muthén (2026) and presented next. The DSEFA approach uses PSEM priors on direct effects from a second-order factor to the factor indicators to arrive at a new type of bi-factor analysis.

## Exploratory Bi-Factor Analysis using Direct Effects Second-Order Factor Analysis (DSEFA)

- Extends SEFA by letting the second-order factor have direct effects on the factor indicators where the effects have ALF priors
- Same model fit as regular EFA with 1 more factor than first-order SEFA factors
- First-order factor loadings have GEOMIN priors as with SEFA
- Significant direct effects suggest the need for DSEFA
- General factor loadings are computed as a sum of indirect and direct effects of the second-order factor on the indicators
- Avoids failures of ROTATION = BI-GEOMIN which are possible in cases with few indicators per factor and no indicators loading mostly on the general factor
- Pre-standardization of indicators important with widely varying variances so that the priors are used optimally
- Available also for categorical indicators using WLSMV estimation

Slide 34 shows the features of DSEFA.

It extends SEFA by letting the second-order factor have direct effects on the factor indicators where the effects have ALF priors.

It has the same model fit as regular EFA with 1 more factor than the number of first-order SEFA factors.

First-order factor loadings have GEOMIN priors as with SEFA.

Significant direct effects suggest the need for DSEFA.

General factor loadings are computed as a sum of indirect and direct effects of the second-order factor on the indicators.

DSEFA avoids failures of ROTATION = BI-GEOMIN which are possible in cases with few indicators per factor and no indicators loading mostly on the general factor.

Pre-standardization is important with widely varying indicator variances.

DSEFA is also available for categorical indicators using the WLSMV estimator.

## Input for Exploratory Bi-Factor Analysis Using DSEFA: Direct Effects Second-Order Factor Analysis

```

ANALYSIS:          ESTIMATOR = MLR;
                   ITERATIONS = 10000;
                   CONVERGENCE = 0.000001;
                   !STARTS = 50;

MODEL:             ! first 3 lines below same as SEFA:
                   f1-f4 BY visual-figurew*(a1-a76); ! 4*19=76
                   f1-f4 (v1-v4);
                   f0 BY f1-f4* (11-14);
                   ! direct effects from the second-order
                   ! factor to the factor indicators:
                   f0 BY visual-figurew*0 (d1-d19);
                   f0@1;

MODEL CONSTRAINT:  v1 = 1 - 11*11;
                   v2 = 1 - 12*12;
                   v3 = 1 - 13*13;
                   v4 = 1 - 14*14;

MODEL PRIORS:      a1-a76 ~ GEOMIN(4,0.1);
                   ! impose alignment function (ALF) priors:
                   d1-d19 ~ ALF(0,1);

```

Slide 35 shows the input for DSEFA. The ANALYSIS command includes settings for iterations, convergence, and starts that may be needed in the manual approach but not in the simplified, automatic approach given on the next slide.

In the MODEL command, the first 3 lines are the same as for SEFA. The specification of the second-order factor is bolded. After this, the direct effects from the second-order factor to the indicators are specified using the BY option (also bolded). Labels that are given for these effects (loadings) and used in the MODEL PRIORS command for the ALF priors.

The ALF priors have mean 0 and variance 1. Here, variance actually is 1/weight where the weight refers to the amount of penalty of the penalized maximum-likelihood estimation. The smaller the prior variance, the further the direct effects are pushed toward their means of zero. For a very small prior variance, it is as if there are no direct effects and the model is a SEFA. A variance of 1 is often a suitable choice. If necessary, it should be varied until the fit of the DSEFA is the same as for EFA with the same number of factors (here 5). For factor indicators that have widely varying variances, standardization of the indicators using the DEFINE command is recommended for the ALF priors to work well.

## Simplified DSEFA Input

- Automated setup of model and priors
  - Special rotation
  - ESEM specification

```
ANALYSIS: ESTIMATOR = MLR;
           ROTATION = DSEFA;
           ! DSEFA settings shown at the
           ! end of the Theory section
```

```
MODEL:    fg f1-f4 by visual-figurew(*1);
           ! covariates can be added
```

- Pre-standardization of factor indicators recommended
- EFA-generated starting values
- DSEFA-specific output
- Current limitations
  - Single group, at least 3 first-order factors, single EFA block

Slide 36 shows the simplified, automated DSEFA input. DSEFA is initiated by a special rotation called DSEFA.

The MODEL command uses an ESEM style specification of the factors where the first factor is the general one.

Pre-standardization of the factor indicators is recommended.

EFA-generated starting values are provided so the special ANALYSIS settings mentioned on the previous slide are typically not needed.

The DSEFA output includes the total effects of the second-order factor on the indicators, computed as the sum of the indirect effects and the direct effects. See also the EFA Theory section on the SEFA - DSEFA transition.

Currently DSEFA is limited to 1 group. At least 3 first-order indicators are needed. And, there cannot be more than one EFA block.

## Results for Exploratory Bi-Factor Results Using DSEFA

### STANDARDIZED SOLUTION

BI-FACTOR MODEL					
	General	Spatial	Verbal	Speed	Memory
VISUAL	<b>0.432*</b>	<b>0.528*</b>	<b>0.185*</b>	0.013	0.062
CUBES	<b>0.267*</b>	<b>0.397*</b>	0.062	-0.119	-0.023
PAPER	<b>0.222*</b>	<b>0.420*</b>	<b>0.151*</b>	0.012	-0.034
FLAGS	<b>0.284*</b>	<b>0.585*</b>	0.005	0.027	0.131
GENERAL	<b>0.345*</b>	0.031	<b>0.767*</b>	0.082	-0.027
PARAGRAPH	<b>0.428*</b>	0.016	<b>0.695*</b>	0.025	0.087
SENTENCE	<b>0.507*</b>	-0.116	<b>0.728*</b>	-0.018	-0.068
WORDC	<b>0.476*</b>	0.031	<b>0.574*</b>	0.014	0.039
WORDM	<b>0.374*</b>	0.110	<b>0.759*</b>	0.014	0.073
ADDITION	0.121	-0.024	0.087	<b>0.771*</b>	0.130
CODE	<b>0.508*</b>	0.033	0.105	<b>0.453*</b>	0.138
COUNTING	<b>0.283*</b>	0.180	0.012	<b>0.584*</b>	-0.024
STRAIGHT	<b>0.586*</b>	0.262	-0.062	<b>0.384*</b>	-0.081
WORDR	0.337	-0.060	0.018	-0.045	<b>0.580*</b>
NUMBERR	0.118	0.157	-0.041	0.023	<b>0.577*</b>
FIGURER	0.416	0.259	0.039	-0.003	<b>0.403*</b>
OBJECT	<b>0.222*</b>	-0.027	-0.009	<b>0.310*</b>	<b>0.509*</b>
NUMBERF	0.222	0.167	0.064	0.182	<b>0.404*</b>
FIGUREW	<b>0.354*</b>	0.044	0.113	0.021	0.261

- All factors are uncorrelated
- The 4 specific factors are well recovered
- ROTATION = DSEFA has the same fit as regular 5-factor EFA and ROTATION = BI-GEOMIN. It is just a different rotation

Slide 37 shows the DSEFA estimates. This output is presented under the headings STANDARDIZED SOLUTION, BI-FACTOR MODEL. All the factors are uncorrelated.

The general factor has mostly significant loadings. The 4 specific factors Spatial, Verbal, Speed, and Memory all have large significant loadings in expected places.

ROTATION = DSEFA has the same fit as regular 5-factor EFA and ROTATION = BI-GEOMIN. It is just a different rotation.

## Comparison with Exploratory Bi-Factor Results Using ROTATION = BI-GEOMIN (TYPE = EFA or ESEM)


ROTATED LOADINGS (\* significant at 5% level)

	General	Verbal	Speed?	Memory	?
VISUAL	0.619*	0.055	-0.339*	-0.011	-0.016
CUBES	0.340*	-0.021	-0.371*	-0.044	0.019
PAPER	0.407*	0.046	-0.249	-0.123	-0.081
FLAGS	0.544*	-0.143	-0.332*	0.021	-0.114
GENERAL	0.420*	<b>0.748*</b>	0.057	-0.098	-0.053
PARAGRAPH	0.436*	<b>0.687*</b>	0.001	0.064	0.024
SENTENCE	0.367*	<b>0.763*</b>	-0.002	-0.013	0.191
WORDC	0.438*	<b>0.569*</b>	-0.046	0.049	0.103
WORDM	0.464*	<b>0.730*</b>	-0.034	-0.005	-0.095
ADDITION	0.476*	-0.003	0.634*	-0.029	-0.045
CODE	0.595*	0.054	0.259*	0.131	0.222
COUNTING	0.564*	-0.105	0.309*	-0.130	0.080
STRAIGHT	0.670*	-0.168	0.011	-0.063	0.335
WORDR	0.247*	0.057	-0.002	<b>0.629*</b>	0.053
NUMBERR	0.263*	-0.073	-0.010	<b>0.493*</b>	-0.209
FIGURER	0.476*	-0.011	-0.178*	<b>0.388*</b>	0.016
OBJECT	0.355*	-0.029	0.287*	<b>0.450*</b>	-0.042
NUMBERF	0.407*	0.016	0.082	<b>0.308*</b>	-0.143
FIGUREW	0.321*	0.114	-0.038	<b>0.285*</b>	0.081

- Spatial and speed factors not recovered: Failure of ROTATION=BI-GEOMIN
- Same fit as regular 5-factor EFA:  $\text{Chi-2}(86) = 113, p = 0.028$

Slide 38 uses the older approach of ROTATION = BI-GEOMIN (Jennrich & Bentler, 2011, 2012). Although the model fit for ROTATION = BI-GEOMIN is the same as for ROTATION = DSEFA on the previous slide, the solution using ROTATION = BI-GEOMIN does not give the same quality of results. With this rotation, only the Verbal and Memory factors are well recovered, whereas the Spatial factor is not recovered and the Speed factor is poorly recovered.

## Holzinger-Swineford 24-Variable Bi-Factor Model



Slide 39 shows a bi-factor model diagram for the H&S data extended by 5 tests to give 24 factor indicators. The 5 test are at the bottom of the diagram. The diagram shows them loading only on the general factor.

## Input for H&S 24-Variable Bifactor Analysis Using CFA, TYPE=EFA, and ESEM

```
MODEL:    ! CFA
          spatial BY visual-flags*;
          verbal BY general-wordm*;
          speed BY addition-straight*;
          memory BY wordr-figurew*;
          g BY visual-arithmet*;
          spatial-g@1;
          ! uncorrelated factors because of the general factor:
          g WITH spatial-memory @0;
          spatial-memory WITH spatial-memory@0;

ANALYSIS: ! EFA
          ESTIMATOR = ML;
          TYPE = EFA 5 5;
          ROTATION = BI-GEOMIN;

ANALYSIS: ! ESEM
          ESTIMATOR = ML;
          ROTATION = BI-GEOMIN;

MODEL:    fg f1-f4 BY visual-arithmet(*1);
```

Slide 40 shows 3 inputs for these data: CFA, EFA, and ESEM. The EFA and ESEM results using ROTATION = BI-GEOMIN will be compared to DSEFA.

## Input for H&S 24-Variable Bifactor Analysis Using DSEFA

```
ANALYSIS: ESTIMATOR = MLR;  
          ROTATION = DSEFA;  
          ! DSEFA settings shown at the  
          ! end of the Theory section  
  
MODEL:   fg f1-f4 by visual-arithmet(*1);  
          ! covariates can be added
```

Slide 41 shows the DSEFA input for the 24-variable analysis. DSEFA is defined by `ROTATION = DSEFA`. The DSEFA option can have several settings related to the priors. These are described in the EFA Theory section. The MODEL statement uses an ESEM type specification.

## Model Fit for H&S 24-Variable Bifactor Models Grant-White and Pasteur Schools, N = 301, MLR

### SUMMARY OF MODEL FIT INFORMATION

Model	Number of Parameters	Chi-Square	Degrees of Freedom	P-Value
4-factor	138	283.322	186	0.0000
5-factor	158	238.592	166	0.0002

Models Compared	Chi-Square	Degrees of Freedom	P-Value
4-factor against 5-factor	38.307	20	0.0081

Slide 42 shows EFA model test results for the 24-variable analysis. In contrast to the 19-variable case, the fit is not good for 4 factors and the 5-factor solution is chosen here. For the bi-factor analyses, a 5-factor solution corresponds to one general and 4 specific factors.

The fit of the 5-factor model can be improved by searching for significant residual covariances. With PSEM, modification indices are not available to guide this search. A simple approach is to instead use an ESEM EFA analysis with 5 factors requesting modindices since the fit of this model is the same as for the DSEFA. This pointed to a significant residual covariance between figurew and addition. When this was freed in the DSEFA analysis, a better model fit was obtained:  $\text{chi-square}(165) = 202$  ( $p = 0.025$ ).

## Findings from Second-Order and Bi-factor Models using the 24-Variable H&S Data

- Competing models:
  - SEFA using 4 first-order factors with the same fit as regular 4-factor EFA
  - DSEFA using 4 first-order factors and a second-order factor with direct effects on the indicators with the same fit as regular 5-factor EFA
- DSEFA shows significant direct effects from the second-order factor to 3 of the 5 indicators added in the 24-variable version which suggests that SEFA is not sufficient but DSEFA needed (compare 4- vs 5-factor testing)
- In the 24-variable case, ROTATION = BI-FACTOR and ROTATION = DSEFA give similar and interpretable solutions, perhaps due to the 5 extra indicators loading mainly on the general factor. This agreement is in line with Asparouhov-Muthén (2026) simulations
- For more on the relationship between SEFA and DSEFA, see the EFA theory section slide SEFA-DSEFA transition

Slide 43 discusses results from second-order and bi-factor analysis for the 24-variable H&S data.

We consider two competing models:

- SEFA using 4 first-order factors with the same fit as regular 4-factor EFA
- DSEFA using 4 first-order factors and a second-order factor with direct effects on the indicators with the same fit as regular 5-factor EFA

DSEFA shows significant direct effects from the second-order factor to 3 of the 5 indicators added in the 24-variable version, which suggests that SEFA is not sufficient but DSEFA is needed (compare 4- vs 5-factor EFA testing).

In the 24-variable case, ROTATION = BI-FACTOR and ROTATION = DSEFA give similar and interpretable solutions, perhaps due to the 5 extra indicators loading mainly on the general factor. The agreement in this case is in line with Asparouhov-Muthén (2026) simulations.

Slide 135 in the EFA theory section says a bit about the interpretation of the different factor loading matrices for SEFA and DSEFA and how the models relate to each other.

## DSEFA Solution for 24-Variable H&S

### STANDARDIZED SOLUTION

#### BI-FACTOR MODEL

	General	Spatial	Verbal	Speed	Memory
VISUAL	<b>0.415*</b>	<b>0.531*</b>	<b>0.182*</b>	0.109	0.046
CUBES	<b>0.274*</b>	<b>0.424*</b>	0.044	-0.064	-0.047
PAPER	<b>0.251*</b>	<b>0.370*</b>	0.112	0.075	-0.069
FLAGS	<b>0.367*</b>	<b>0.535*</b>	-0.057	0.086	0.070
GENERAL	<b>0.398*</b>	-0.009	<b>0.728*</b>	0.040	-0.068
PARAGRAPH	<b>0.442*</b>	0.026	<b>0.690*</b>	0.005	0.062
SENTENCE	<b>0.360*</b>	0.007	<b>0.808*</b>	0.012	-0.025
WORDC	<b>0.427*</b>	0.077	<b>0.602*</b>	0.022	0.039
WORDM	<b>0.474*</b>	0.068	<b>0.695*</b>	-0.040	0.012
ADDITION	0.424	-0.317	-0.052	<b>0.609*</b>	0.003
CODE	<b>0.370*</b>	0.040	<b>0.206*</b>	<b>0.525*</b>	<b>0.184*</b>
COUNTING	<b>0.378*</b>	0.039	-0.023	<b>0.540*</b>	-0.062
STRAIGHT	<b>0.348*</b>	<b>0.307*</b>	0.079	<b>0.522*</b>	0.006
WORDR	<b>0.281*</b>	-0.005	0.067	-0.025	<b>0.596*</b>
NUMBERR	<b>0.233*</b>	0.105	-0.103	0.006	<b>0.516*</b>
FIGURER	<b>0.462*</b>	<b>0.248*</b>	0.017	0.012	<b>0.369*</b>
OBJECT	<b>0.254*</b>	-0.054	-0.004	<b>0.287*</b>	<b>0.500*</b>
NUMBERF	<b>0.285*</b>	0.087	0.037	<b>0.180*</b>	<b>0.380*</b>
FIGUREW	<b>0.408*</b>	0.046	0.101	-0.017	<b>0.237*</b>
DEDUCT	<b>0.521*</b>	<b>0.266*</b>	0.155	<b>-0.137*</b>	0.102
NUMERIC	<b>0.664*</b>	0.111	0.048	0.140	-0.025
PROBLEMR	<b>0.529*</b>	0.233	<b>0.320*</b>	-0.057	0.018
SERIES	<b>0.674*</b>	<b>0.278*</b>	<b>0.179*</b>	-0.021	-0.001
ARITHMET	<b>0.689*</b>	-0.121	0.172	0.088	0.056

- The 5 extra tests load mainly on the general factor

Slide 44 shows the DSEFA solution for the 24-variable case. This output is presented under the headings STANDARDIZED SOLUTION, BI-FACTOR MODEL. All the factors are uncorrelated.

The 4 specific factors are all well recovered. The general factor has significant loadings for all variables. The 5 extra variables mainly load on the general factor with some cross loadings.